Shader "UI/AlphaGradient"
{
    Properties
    {
        [PerRendererData] _MainTex ("Texture", 2D) = "white" {}
        _Color ("Tint", Color) = (1,1,1,1)
        
        _StencilComp ("Stencil Comparison", Float) = 8
        _Stencil ("Stencil ID", Float) = 0
        _StencilOp ("Stencil Operation", Float) = 0
        _StencilWriteMask ("Stencil Write Mask", Float) = 255
        _StencilReadMask ("Stencil Read Mask", Float) = 255
        
        _ColorMask ("Color Mask", Float) = 15
        
        [Toggle(UNITY_UI_ALPHACLIP)] _UseUIAlphaClip ("Use Alpha Clip", Float) = 0
        
        [Header(Alpha Gradient)]
        _GradientValue("Gradient Value (x:Down,y:Up,z:Left,w:Right)",vector) = (0,0,0,0)
        _GradientSmooth("Gradient Smooth", Range(0.01,1)) = 0.5
        
        [Header(Alpha Gradient 2)]
        _GradientValue2("Gradient Value2 (x:Down,y:Up,z:Left,w:Right)",vector) = (0,0,0,0)
        _GradientSmooth2("Gradient Smooth2", Range(0.01,1)) = 0.01
        
        [Toggle]_ToGray("ToGray",float) = 0
    }
    SubShader
    {
        Tags
        {
            "Queue"="Transparent"
            "IgnoreProjector"="True"
            "RenderType"="Transparent"
            "PreviewType"="Plane"
            "CanUseSpriteAtlas"="True"
        }
        
        Stencil
        {
            Ref [_Stencil]
            Comp [_StencilComp]
            Pass [_StencilOp]
            ReadMask [_StencilReadMask]
            WriteMask [_StencilWriteMask]
        }

        Cull Off
        Lighting Off
        ZWrite Off
        ZTest [unity_GUIZTestMode]
        Blend SrcAlpha OneMinusSrcAlpha
        ColorMask [_ColorMask]
        
        Pass
        {
            Name "Default"
            
            CGPROGRAM
            
            #pragma vertex vert
            #pragma fragment frag
            #pragma target 2.0

            #include "UnityCG.cginc"
            #include "UnityUI.cginc"

            #pragma multi_compile_local _ UNITY_UI_CLIP_RECT
            #pragma multi_compile_local _ UNITY_UI_ALPHACLIP

            struct appdata_t
            {
                float4 vertex   : POSITION;
                float4 color    : COLOR;
                float2 texcoord : TEXCOORD0;
                UNITY_VERTEX_INPUT_INSTANCE_ID
            };

            struct v2f
            {
                float4 vertex   : SV_POSITION;
                float4 color    : COLOR;
                float2 texcoord  : TEXCOORD0;
                float4 worldPosition : TEXCOORD1;
                UNITY_VERTEX_OUTPUT_STEREO
            };

            sampler2D _MainTex;
            float4 _Color;
            float4 _TextureSampleAdd;
            float4 _ClipRect;
            float4 _MainTex_ST;
            float _GradientSmooth;
            float _GradientSmooth2;

            float4 _GradientValue;
            float4 _GradientValue2;
            float _ToGray;

            v2f vert(appdata_t v)
            {
                v2f OUT;
                
                UNITY_SETUP_INSTANCE_ID(v);
                UNITY_INITIALIZE_VERTEX_OUTPUT_STEREO(OUT);
                
                OUT.worldPosition = v.vertex;
                OUT.vertex = UnityObjectToClipPos(OUT.worldPosition);
                OUT.texcoord = TRANSFORM_TEX(v.texcoord, _MainTex);
                OUT.color = v.color * _Color;
                
                return OUT;
            }

            float4 frag(v2f IN) : SV_Target
            {
                half4 color = (tex2D(_MainTex, IN.texcoord) + _TextureSampleAdd) * IN.color;

                #ifdef UNITY_UI_CLIP_RECT
                color.a *= UnityGet2DClipping(IN.worldPosition.xy, _ClipRect);
                #endif

                #ifdef UNITY_UI_ALPHACLIP
                clip (color.a - 0.001);
                #endif

                float alpha = color.a - smoothstep(IN.texcoord.y , IN.texcoord.y + _GradientSmooth , _GradientValue.x);
                alpha -= smoothstep( 1 - IN.texcoord.y , 1 - IN.texcoord.y + _GradientSmooth, _GradientValue.y);
                alpha -= smoothstep( IN.texcoord.x , IN.texcoord.x + _GradientSmooth, _GradientValue.w);
                alpha -= smoothstep( 1 - IN.texcoord.x , 1 - IN.texcoord.x + _GradientSmooth, _GradientValue.z);
                alpha = saturate(alpha);

                float alpha2 = color.a - smoothstep(IN.texcoord.y , IN.texcoord.y + _GradientSmooth2 , _GradientValue2.x);
                alpha2 -= smoothstep( 1 - IN.texcoord.y , 1 - IN.texcoord.y + _GradientSmooth2, _GradientValue2.y);
                alpha2 -= smoothstep( IN.texcoord.x , IN.texcoord.x + _GradientSmooth2, _GradientValue2.w);
                alpha2 -= smoothstep( 1 - IN.texcoord.x , 1 - IN.texcoord.x + _GradientSmooth2, _GradientValue2.z);
                alpha2 = saturate(alpha2);

                color.a = alpha * alpha2;

                color.rgb = lerp(color.rgb , Luminance(color.rgb) ,_ToGray);
                
                return color;
            }
            
            ENDCG
            
        }
    }
}
